

As aulas de 1 a 5 foram elaboradas juntamente com o Prof. Ma To FU (UEM)

NOÇÕES DE CÁLCULO DIFERENCIAL E INTEGRAL

com uma variável

> **# Simbolo para comentário**

1 - CALCULANDO LIMITES

> **# assim podemos definir uma função**

> **f1:=(x,y) -> (x^2-5*y)/(x^3+2*x);**

$$f1 := (x, y) \to \frac{x^2 - 5y}{x^3 + 2x}$$

> **limit(f1(x,y),x=1);**

$$\frac{1}{3} - \frac{5}{3}y$$

Vamos ver o gráfico de uma função.

> **plot((x^2-5)/(x^3+2*x),x=-2..2);**



Outra forma de definir uma função é por meio do chamado procedimento.

Um procedimento tem a forma:

P : = proc(argumentos)

local (variáveis locais)

instruções a serem executadas

end;

Vamos ver um exemplo, mais tarde voltaremos neste ponto com mais cuidado.

```
> salto:=proc(x);
> if 1<=x then x^2+1 else cos(x) fi;
> end;
```

$$salto := \text{proc}(x) \text{ if } 1 \leq x \text{ then } x\text{\textasciicircum}2 + 1 \text{ else } \cos(x) \text{ fi end}$$

```
> salto(-1);
```

$$\cos(1)$$

```
> salto(2);
```

$$5$$

Outro exemplo.

```
> pp:=proc(x);
```

```
> x^3;
> end;
```

$$pp := \text{proc}(x)\ x\text{^}3\ \text{end}$$

```
> pp(sqrt(2));
```

$$2\sqrt{2}$$

```
> pp((2)^(1/3));
```

$$2$$

```
> # Agora vamos voltar aos limites e escrever Limit no lugar de limit.
> #
> Limit(f1(x,2), x=1);
```

$$\lim_{x \to 1} \frac{x^2 - 10}{x^3 + 2x}$$

```
> value(");
```

$$-3$$

Se um comando começa com letra maiúscula, então ele é "inerte". Isto é, só escreve mas não calcula.

```
> A:=sin(2*x)/x;
```

$$A := \frac{\sin(2x)}{x}$$

Observar que A não é do tipo A:=x -> sin(2*x)/x.

Portanto A não é função (é uma expressão).

```
> Limit(A, x=0);
```

$$\lim_{x \to 0} \frac{\sin(2x)}{x}$$

```
> value(");
```

$$2$$

```
> Limit(A, x=infinity);
```

$$\lim_{x \to \infty} \frac{\sin(2x)}{x}$$

```
> value(");
```

$$0$$

Lá vão dois limites infinitos.

```
> Limit(1/x, x=0, left);
```

$$\lim_{x \to 0-} \frac{1}{x}$$

```
> value(");
```

$$-\infty$$

Abaixo, usamos "limit" com l minúscula.

```
> limit(1/x, x=0, right);
```

$$\infty$$

2 - CALCULANDO ALGUMAS DERIVADAS

```
> f2:=x -> sin(2*x);
```

$$f2 := x \to \sin(2\,x)$$

```
> # Derivando uma vez.
> diff(f2(x),x);
```

$$2\cos(2\,x)$$

```
> # Derivando 3 vezes.
> diff(f2(x),x,x,x);
```

$$-8\cos(2\,x)$$

```
> #
> # Observa que para derivar 3 vezes podemos usar x,x,x; ou $3.
> diff(f2(x),x$3);
```

$$-8\cos(2\,x)$$

```
> #
```

Calculando derivadas com operador diferencial "D" A saída (output) é sempre uma função

```
> h:=x -> x^2;
```

$$h := x \to x^2$$

> h(x);

$$x^2$$

> D(h); # Vai sair uma função !

$$x \to 2\,x$$

> D(h)(x);

$$2\,x$$

> diff(f(x)*g(x),x); # derivada do produto

$$\left(\frac{\partial}{\partial x}\mathrm{f}(x)\right)\mathrm{g}(x) + \mathrm{f}(x)\left(\frac{\partial}{\partial x}\mathrm{g}(x)\right)$$

> diff(f(x,y(x)) , x); # derivada total de f(x,y(x))

$$\mathrm{D}_1(f)(x, \mathrm{y}(x)) + \mathrm{D}_2(f)(x, \mathrm{y}(x))\left(\frac{\partial}{\partial x}\mathrm{y}(x)\right)$$

> #

3 - UM PROBLEMINHA DE CÁLCULO INTERESSANTE PARA DISTRAIR.

Dado um ponto P=(a,b) e o gráfico de uma função f,

calcular a distância de P ao gráfico da f.

Ao começar um novo problema costuma-se "zerar" a

memória.

> restart:

Por definição, a distância de P ao gráfico da f é a menor distância entre os pontos Q do gráfico e o ponto P. Vamos então definir:

distância=|| P-Q ||

Neste exemplo tomamos f(x)=cos(x) e P=(2,3).

> dist:=sqrt((2-x)^2 + (3-cos(x))^2);

$$dist := \sqrt{(2 - x)^2 + (3 - \cos(x))^2}$$

Queremos a menor distância. Da fórmula acima vemos que dist é uma função só de x. Portanto, do Cálculo, concluímos que a menor distância deve ser dada por

um x que anula a derivada de dist.

Vamos ver o gráfico da função dist.

> **plot(sqrt((2-x)^2 + (3-cos(x))^2),x=0..2);**



> **derivada:=diff(dist,x);**

$$derivada := \frac{1}{2}\frac{2\,x - 4 + 2\,(3 - \cos(x))\,\sin(x)}{\sqrt{(2 - x)^2 + (3 - \cos(x))^2}}$$

Resolvendo a equação derivada=0 para "x" em [0,2]. Já vi o gráfico.

> **x[min]:=fsolve(derivada, x ,0..2);**

$$x_{min} := .6551969516$$

> **evalf(");**

$$.6551969516$$

> **#**

Escrevendo a resposta utilizando o comando subs. (Substituir o valor de x por x_minimo na expressão dist).

> **menor[dist]:=subs(x=x[min], dist);**

$$menor_{\sqrt{(2 - x)^2 + (3 - \cos(x))^2}} := \sqrt{1.808495238 + (3 - \cos(.6551969516))^2}$$

> **resposta:=evalf(");**

$$resposta := 2.584504271$$

> # Descubra o que foi feito (abaixo)

> L:=[[2,3], [x[min] , cos(x[min])]];

$$L := [[2, 3], [.6551969516, .7929279378]]$$

> plot({cos(x), L }, x=0..3 , scaling=constrained);



4 - SÉRIES DE TAYLOR

A função "series" escreve a 'série de Taylor de funções analíticas. Em geral, a resposta é dada em termos de uma expansão de ordem 6.

> S:=exp(x)*x^2;

$$S := e^x x^2$$

> S1:=series(S, x=0); # Em torno de x=0.

$$S1 := x^2 + x^3 + \frac{1}{2}x^4 + \frac{1}{6}x^5 + O(x^6)$$

Queremos agora uma expansão de ordem 10.

> S2:=series(S , x=0, 10);

$$S2 := x^2 + x^3 + \frac{1}{2}x^4 + \frac{1}{6}x^5 + \frac{1}{24}x^6 + \frac{1}{120}x^7 + \frac{1}{720}x^8 + \frac{1}{5040}x^9 + O(x^{10})$$

Converter a série num polinômio.

> S3:=convert(S2, polynom);

$$S3 := x^2 + x^3 + \frac{1}{2}x^4 + \frac{1}{6}x^5 + \frac{1}{24}x^6 + \frac{1}{120}x^7 + \frac{1}{720}x^8 + \frac{1}{5040}x^9$$

> S3(2); # inútil

$$\mathrm{x}(2)^2 + \mathrm{x}(2)^3 + \frac{1}{2}\mathrm{x}(2)^4 + \frac{1}{6}\mathrm{x}(2)^5 + \frac{1}{24}\mathrm{x}(2)^6 + \frac{1}{120}\mathrm{x}(2)^7 + \frac{1}{720}\mathrm{x}(2)^8 + \frac{1}{5040}\mathrm{x}(2)^9$$

> value("); #inútil

$$\mathrm{table}([min = .6551969516])(2)^2 + \mathrm{table}([min = .6551969516])(2)^3 + \frac{1}{2}\mathrm{table}([min = .655196951$$

$$+ \frac{1}{6}\mathrm{table}([min = .6551969516])(2)^5 + \frac{1}{24}\mathrm{table}([min = .6551969516])(2)^6$$

$$+ \frac{1}{120}\mathrm{table}([min = .6551969516])(2)^7 + \frac{1}{720}\mathrm{table}([min = .6551969516])(2)^8$$

$$+ \frac{1}{5040}\mathrm{table}([min = .6551969516])(2)^9$$

Vamos transformar a expressão S3 numa função de verdade.

> P:=unapply(S3 , x);

$$P := x \rightarrow x^2 + x^3 + \frac{1}{2}x^4 + \frac{1}{6}x^5 + \frac{1}{24}x^6 + \frac{1}{120}x^7 + \frac{1}{720}x^8 + \frac{1}{5040}x^9$$

> evalf(P(1));

$$2.718253968$$

> evalf(subs(x=1, S));

$$2.718281828$$

5 - CALCULANDO INTEGRAIS

> Int(ln(x),x); # Com I maiúscula

$$\int \ln(x)\, dx$$

> int(ln(x),x); # Com i minúscula

$$x \ln(x) - x$$

> int(tan(x),x); # Com I minúscula

$$-\ln(\cos(x))$$

> Int(1/ sqrt(1-x^2),x);

$$\int \frac{1}{\sqrt{1-x^2}}\,dx$$

> value(");

$$\arcsin(x)$$

> diff(",x);

$$\frac{1}{\sqrt{1-x^2}}$$

> AA:=Int(x^2*exp(x^2),x=0..1);

$$AA := \int_0^1 x^2\,e^{(x^2)}\,dx$$

> resp:=value(AA);

$$resp := \frac{1}{2}\,e + \frac{1}{4}\,I\sqrt{\pi}\,\mathrm{erf}(I)$$

> #

Maple utilizou a função erro "erf" que é definida por

erf(x) = 2 / sqrt(Pi) * int(exp(-t^2), t=0..x)

Mas mesmo assim podemos ter uma aproximação da integral utilizando-se evalf(AA).

> evalf(AA);

$$.6278150413$$

Os próximos exemplos são conhecidos.

> #

> # Integrais Múltiplas

> #

> Int(Int(x^2+y^2, y=0..2),x=0..2);

$$\int_0^2\int_0^2 x^2 + y^2\,dy\,dx$$

> value(");

$$\frac{32}{3}$$

```
> Int( Int( Int( z*cos(x), x=0..2*y ), y=0..2 ), z=1..2 );
```

$$\int_1^2 \int_0^2 \int_0^{2y} z \cos(x)\, dx\, dy\, dz$$

```
> value(");
```

$$\frac{3}{2} \sin(2)^2$$

```
> evalf(");
```

$$1.240232716$$

Veremos alguns exemplos de integrais improprias.

```
> f8:= exp(-u*x)*ln(x)*sqrt(x);
```

$$f8 := e^{(-u\,x)} \ln(x) \sqrt{x}$$

```
> int(f8,x=0..infinity);
```

Definite integration: Can't determine if the integral is convergent.
Need to know the sign of --> u
Will now try indefinite integration and then take limits.

$$\int_0^{\infty} e^{(-u\,x)} \ln(x) \sqrt{x}\, dx$$

Como nada é conhecido sobre u, MapleV não pode determinar

uma resposta. Podemos usar o comando ASSUME para informar

ao MapleV sobre u.

```
> assume(u<0); int(f8,x=0..infinity);
```

$$\infty$$

Esta integral diverge. Por outro lado, se u>0 temos convergência e uma resposta simbólica.

```
> assume(u>0);int(f8,x=0..infinity);
```

$$-\frac{1}{2}\frac{\ln(u\sim)\sqrt{\pi}}{u\sim^{3/2}} + \frac{\sqrt{\pi}}{u\sim^{3/2}} - \frac{1}{2}\frac{\sqrt{\pi}\,\gamma}{u\sim^{3/2}} - \frac{\sqrt{\pi}\ln(2)}{u\sim^{3/2}}$$

```
> #
```

```
> Int(1/x^2 , x=1..infinity);
```

$$\int_1^{\infty} \frac{1}{x^2}\, dx$$

```
> value(");
```

$$1$$

```
> #
> # Outro exemplo.
> #
> int(1/x, x=0..1);
```

$$\infty$$

```
> #
> # Observe isto !
> #
> int(1/(x^3), x=a..b);
```

$$-\frac{1}{2}\frac{1}{b^2} + \frac{1}{2}\frac{1}{a^2}$$

```
> subs(a=-1, b=4, " );
```

$$\frac{15}{32}$$

```
> # Ou ainda:
> int(1/(x^3), x=-1..4, 'CauchyPrincipalValue' );
```

$$\frac{15}{32}$$

Um exemplo sobre transformada de Laplace.

```
> restart; with(inttrans):
> laplace(t**3-cos(t)=y(t), t, s);
```

$$\frac{6}{s^4} - \frac{s}{s^2+1} = \mathrm{laplace}(\mathrm{y}(t), t, s)$$

> **laplace(t^(1/2)-exp(-t)+sinh(a*t), t, s);**

$$\frac{1}{2}\frac{\sqrt{\pi}}{s^{3/2}}-\frac{1}{s+1}+\frac{a}{s^2-a^2}$$

>